
Zusammenfassung 

Der Student-Newman-Keuls-Test ist ein Multipler Mittelwertvergleich, der häufig bei 
landwirtschaftlichen Feldversuchen z.B. zur Prüfung von Pflanzenschutzmitteln oder 
Pflanzensorten eingesetzt wird. Dieser Test steht leider nicht standardmäßig in der SAS 
Prozedur MIXED zur Verfügung. Wir zeigen die Berechnung des SNK-Tests basierend auf 
Output der Prozedur MIXED mit Hilfe von selbst programmierten SAS-Makros und unter 
Nutzung der Base SAS-Funktion PROBMC. 

 
Schlüsselwörter: Student-Newman-Keuls, Multipler Mittelwertvergleich, MIXED-
Prozedur, PROBMC-Funktion 

1 Einleitung 
Landwirtschaftliche Experimente sind häufig so angelegt, dass nominalskalierte 
Prüffaktoren (z.B. Pflanzensorten, Pflanzenschutzmittelformulierungen) in mehreren 
Stufen vorliegen und den Versuchseinheiten (Parzellen, Gefäße) randomisiert zugeord-
net werden. Zur Auswertung intervallskalierter Merkmale werden dann varianzanalyti-
sche Verfahren eingesetzt und ein F-Test durchgeführt. Hierfür stehen in SAS die Pro-
zeduren GLM, MIXED und GLIMMIX zur Verfügung. Nach signifikantem Globaltest 
werden die erhaltenen Schätzer für die Erwartungswerte der Faktorstufen verglichen. 
Da fast immer mehr als zwei Faktorstufen geprüft werden, entsteht ein Multiplizitäts-
problem [1]. Der multiple t-Test hält das Niveau für den Fehler erster Art nur für spezi-
fische Einzelvergleiche nicht jedoch in Bezug auf das gesamte Experiment ein. Je mehr 
Mittelwerte am Ende miteinander verglichen werden, desto höher wird das Risiko zu-
fälligerweise signifikante Einzelvergleiche zu finden. Um dieses Risiko kontrollieren zu 
können, wurden von verschiedenen Autoren Adjustierungsverfahren entwickelt. Im 
landwirtschaftlichen Kontext erfreuen sich die Verfahren nach Tukey [10] und Student-
Newman-Keuls (SNK) [9], [4], [2] großer Beliebtheit. Einen Überblick über die Umset-
zung verschiedener Mittelwerttests in SAS geben Schumacher und Weimer [8]. Der 
SNK-Test ist jedoch bislang in Verbindung mit der Prozedur MIXED nicht verfügbar. 

1.1 Der Student-Newman-Keuls-Test 

Der Mittelwertvergleich nach Student-Newman-Keuls [9], [4], [2] prüft sequentiell 
Mittelwertdifferenzen hinsichtlich ihrer Signifikanz. Hierzu werden die Mittelwerte zu-
nächst aufsteigend geordnet und erhalten die Ränge 1 bis m. Die größte Mittelwertdiffe-
renz mit der gleichzeitig größten Rangdifferenz r=m-1 wird als erstes geprüft. Falls 
diese Differenz nicht signifikant ist, bricht das Verfahren ab. Ist die größte Differenz 
jedoch auf dem gewählten Niveau signifikant, so kann der Vergleich des größten gegen 
den zweitkleinsten Mittelwert erfolgen. So wird schrittweise fortgefahren, bis alle mög-
lichen Differenzen geprüft wurden. Zu beachten ist die Regel, wonach Vergleiche nicht 
signifikant sind, wenn gleichzeitig Vergleiche unter Beteiligung von noch kleineren 
oder noch größeren Mittelwerten nicht signifikant sind. Zur Identifikation signifikanter 
Vergleiche in einer Mittelwerttabelle wird üblicherweise eine Buchstabendarstellung 
genutzt. 
Die kritische Differenz zweier Mittelwerte im SNK-Test berechnet sich nach: 

 
 
 
 

Mit α = Signifikanzniveau, k = Anzahl vom Vergleich überspannter Mittelwerte, v = 
Fehlerfreiheitsgrade, MSE = Fehlervarianz, n = Anzahl Wiederholungen. Die Werte für 
q sind in Statistik-Lehrbüchern tabelliert und stehen auch innerhalb von Statistiksoft-
ware wie R oder SAS zur Verfügung. 
Beispiel: In einem Experiment wurden 14 Varianten in vier Blocks getestet. Es stehen 
56-1-13-3 = 39 Fehlerfreiheitsgrade zur Verfügung. Die Fehlervarianz wurde auf 4.21 
geschätzt. Den kleinsten Mittelwert erreichte Variante #1 mit 48.21. Den viertkleinsten 
Mittelwert erreichte die Variante #10 mit durchschnittlich 81.00. Ist die Mittelwertdiffe-
renz von 81.00 – 48.21 = 32.79 signifikant? Da der Vergleich insgesamt 4 Mittelwerte 
überspannt, benötigen wir q(0.05; 4; 39). In einer q-Tabelle finden wir für α=0.05, 40 
Freiheitsgrade den Wert q=3.79, für 30 Freiheitsgrade den Wert 3.85. Wir interpolieren 
und wählen q=3.80. Damit ergibt sich: 

 
 
 
 

Da die gemessene Differenz deutlich größer als 3.90 ist können wir schließen, dass die 
Mittelwertdifferenz auf Niveau α=0.05 signifikant ist. In gleicher Form werden die kri-
tischen Mittelwertdifferenzen für alle anderen Vergleiche berechnet. Ein solcher Test 
kann in SAS mit der Prozedur GLM und der Option SNK im MEANS-Statement ange-
wiesen werden. 
  Proc glm data = Daten; 
  class WDH F1; 
  model Ertrag = Wdh F1 /ss3; 
  means F1 / SNK; 
  run; 



Poster 

85 

Tabelle 1: Output SAS Prozedur GLM, SNK-Test mit Buchstabendarstellung 

Alpha 0.05

Error Degrees of Freedom 39

Error Mean Square 4.21

 

Number of Means 2 3 4 5 6 7 8 9

Critical Range 2.9357 3.5360 3.8946 4.1502 4.348 4.509 4.6457 4.7630

 

Number of Means 10 11 12 13 14

Critical Range 4.8660 4.9578 5.0405 5.1157 5.1845

 

Means with the same letter are not significantly different. 

SNK Grouping Mean N Variante 

 A 90.658 4 7 

 A 90.333 4 6 

 A 89.565 4 8 

 A 89.480 4 5 

 A 89.383 4 13 

 A 89.355 4 3 

 A 89.338 4 4 

 A 89.153 4 2 

B A 86.015 4 12 

B  84.885 4 11 

 C 81.000 4 10 

 C 80.155 4 9 

 D 76.943 4 14 

 E 48.213 4 1 

 
Kritisch ist beim SNK-Test zu sehen, dass er den versuchsweisen Fehler 1. Art nicht 
einhält und beim vergleichsbezogenen Fehler 2. Art zu konservativ ist. Er ist als Kom-
promiss zwischen Tukey-Test und multiplem t-Test zu sehen. Der SNK-Test stellt je-
doch in Europa für die Pflanzenschutzmittelindustrie und Behörden, die mit der Prüfung 
von Pflanzenschutzmitteln beauftragt sind, nach wie vor einen Standard dar. 

1.2 Gemischte Modelle und die SAS Prozedur MIXED 

In den letzten Jahren ist man bei der Auswertung landwirtschaftlicher Feldversuche zu-
nehmend zur Verwendung Linearer Gemischter Modelle übergegangen. Dieses bietet 
Vorteile für bestimmte Versuchsdesigns mit mehr als einem Fehlerterm wie z.B. Spalt-
anlagen und Streifenanlagen, für die Auswertung von Versuchsserien und auch im Falle 
von unvollständigen Blocks oder fehlenden Werten. 
In SAS steht zur Auswertung von Gemischten Modellen die Prozedur MIXED zur Ver-
fügung. Auch für die Auswertung von Pflanzenschutzmittelprüfungen werden in zu-
nehmendem Maße Gemischte Modelle und Generalisierte Gemischte Modelle (z.B. mit 
der Prozedur GLIMMIX) eingesetzt. 
Während wie oben gezeigt die SAS Prozedur GLM mit der Option „SNK“ eine auto-
matische p-Wert Adjustierung und Buchstabendarstellung zur Identifizierung signifi-
kanter Vergleiche anbietet, sucht man eine solche Option bei PROC MIXED vergeb-
lich. Auch PROC GLIMMIX bietet den SNK-Test nicht an. Die Nichtverfügbarkeit des 
SNK-Tests stellt deshalb ein gewisses Hindernis für den Einsatz der MIXED Prozedur 
bei der Auswertung von Pflanzenschutzmittelversuchen dar. 

1.3 PIAF und PIAFStat 

Für viele Versuchsansteller und Behörden ist das Software-System PIAF (Planungs-, 
Informations- und Auswertungssystem für Feldversuche) mittlerweile eine wichtige 
Plattform für die Statistische Datenanalyse [7]. Hierbei wird SAS-Code zentral über die 
Programmbibliothek PIAFStat bereitgestellt [3]. Auch hier wird zunehmend die 
MIXED-Prozedur eingesetzt. 
Der Fokus bei PIAFStat lag bislang auf Sortenversuchen. Im Pflanzenschutzbereich 
stand bisher nur die von der proPlant GmbH in Münster innerhalb des PIAF-Systems 
programmierte Varianzanalyse mit dem SNK-Test zur Verfügung. Viele Eigenschaften 
von landwirtschaftlichen Versuchen wie unvollständige Blöcke, fehlende Werte oder 
mehrere Fehlerterme können von dem PIAF-PSM Verfahren bislang nicht adäquat ab-
gebildet werden. Um die vielfältigen Verfahrensentwicklungen aus dem PIAFStat-Um-
feld auch für Pflanzenschutzversuche nutzen zu können, wird daher angestrebt, die 
beiden Systeme enger zusammenzuführen. Die Möglichkeit den SNK-Test auch für mit 
PROC MIXED programmierte Verfahren zur Verfügung zu haben, ist hierbei ein ganz 
wichtiges Kriterium. 
 
Auch außerhalb der PIAFStat-Umgebung ist die Möglichkeit, den SNK-Test mit Ge-
mischten Modellen kombinieren zu können, von hoher Bedeutung. So wird PROC 
MIXED zunehmend auch in der BASF SE bei der Pflanzenschutzmittelentwicklung für 
die Auswertung von Feldversuchen eingesetzt. 
 
Die in diesem Beitrag vorgestellten SAS Makros wurden in Kooperation zwischen der 
Arbeitsgruppe „PIAF Auswertung“, der Landesforschungsanstalt für Landwirtschaft 
und Fischerei Mecklenburg-Vorpommern und Statistikern der BASF SE entwickelt. 

2 Programmablauf SNK-Test mit PROC MIXED 
Wir haben in SAS und damit in einem PIAFStat-Verfahren ‚A-Bl SNK‘  folgende Lö-
sung unter Nutzung mehrerer zum Teil selbst programmierter SAS Makros umgesetzt: 

 Zunächst erfolgt ein normaler Aufruf von PROC MIXED. Prüfgliedmittelwerte, 
Prüfglieddifferenzen und Überschreitungswahrscheinlichkeiten (p-Werte) dieser 
Differenzen werden mittels ODS in zwei Ausgabedateien (lsmeans, diffs) ge-
schrieben. 

 Für die Adjustierung nach dem SNK-Verfahren nutzen wir die SAS Funktion 
„probmc“. Die Mittelwerte werden ihrer Größe nach geordnet, und neben der 
Mittelwertdifferenz wird auch die Rangdifferenz ermittelt. Je nach Anzahl über-
spannter Rangplätze erfolgt dann eine Adjustierung der p-Werte entsprechend 
studentisierter Spannweiten. 

 Die Dateien mit den Mittelwerten und Prüfglieddifferenzen nebst nach SNK 
adjustierten p-Werten werden einem von Piepho [5], [6] entwickelten und von 
uns modifizierten SAS-Makro (%mult) übergeben. Das %mult-Makro nimmt 
diese Informationen auf und produziert die gewünschte Buchstabendarstellung. 

 Da die Buchstabendarstellung beim SNK-Verfahren zu Redundanzen führen 
kann, erfolgt abschließend in einem weiteren Makro ein Überarbeiten der Buch-
staben. 

3 SAS-Code im Detail 

3.1 Berechnen von Mittelwerten und Mittelwertdifferenzen (MIXED) 

Die Daten werden zunächst mit der Prozedur Mixed ausgewertet. Neben F-Tests und 
Varianzkomponentenschätzung werden auch Mittelwerte und Differenzen berechnet. 
Diese werden per ODS in zwei Dateien geschrieben. In diesem Beispiel haben die Da-
teien die Namen „lsmeans_raw“ und „diffs_raw“. 
 
Code für einen einfaktoriellen Einzelversuch (Randomisierte Blockanlage): 
proc mixed data=daten; 
  class F1 Wdh; 
  model ERTRAG = F1 Wdh / ddfm=kr; 
  lsmeans F1  / pdiff cl; 
ods output lsmeans=lsmeans_raw diffs=diffs_raw; 
run; 

Tabelle 2: Mittelwerttabelle „lsmeans_raw“ mit Konfidenzintervallen 

F1 Estimate Stand. Error DF tValue Pr > |t| Alpha Lower Upper
1 48.21 1.18 24 40.99 <.0001 0.05 45.79 50.64
2 89.15 1.18 24 75.79 <.0001 0.05 86.73 91.58
3 89.36 1.18 24 75.97 <.0001 0.05 86.93 91.78
4 89.34 1.18 24 75.95 <.0001 0.05 86.91 91.76
5 89.48 1.18 24 76.07 <.0001 0.05 87.05 91.91
6 90.33 1.18 24 76.8 <.0001 0.05 87.91 92.76
7 90.66 1.18 24 77.07 <.0001 0.05 88.23 93.08
8 89.57 1.18 24 76.14 <.0001 0.05 87.14 91.99
9 80.16 1.18 24 68.14 <.0001 0.05 77.73 82.58
10 81.00 1.18 24 68.86 <.0001 0.05 78.57 83.43
11 84.89 1.18 24 72.17 <.0001 0.05 82.46 87.31
12 86.02 1.18 24 73.13 <.0001 0.05 83.59 88.44
13 89.38 1.18 24 75.99 <.0001 0.05 86.96 91.81
14 76.94 1.18 24 65.41 <.0001 0.05 74.52 79.37
 
Tabelle 3: Mittelwertdifferenzen, „diffs_raw“ 

F1 _F1 Estimate Stand. Error DF tValue Pr > |t| Alpha Lower Upper
1 2 -40.94 1.45 39 -28.21 <.0001 0.05 -43.88 -38.00
1 3 -41.14 1.45 39 -28.35 <.0001 0.05 -44.08 -38.21
1 4 -41.13 1.45 39 -28.33 <.0001 0.05 -44.06 -38.19
... ... ... ... ... ... ... ... ... ...
11 14 7.94 1.45 39 5.47 <.0001 0.05 5.01 10.88
12 13 -3.37 1.45 39 -2.32 0.0257 0.05 -6.30 -0.43
12 14 9.07 1.45 39 6.25 <.0001 0.05 6.14 12.01
13 14 12.44 1.45 39 8.57 <.0001 0.05 9.50 15.38

3.2 Berechnen adjustierter Konfidenzintervalle (%critical_range_SNK) 

Das Makro %critical_range_SNK nimmt die SAS-Datei „diffs_raw“ auf und ergänzt die 
kritischen Differenzen entsprechend dem SNK-Verfahren.  

 Zunächst werden die Mittelwerte sortiert und mit Rangzahlen versehen.  
 Im zweiten Schritt kombinieren wir die Ränge mit der Datei „diffs_raw“ und be-

rechnen die Rangdifferenz zweier Mittelwerte. Für jeden paarweisen Vergleich 
kann so ermittelt werden, wie viele Ränge der Vergleich überspannt. Dieser Wert 
wird als Variable „Range“ in der Tabelle abgespeichert. 

 Im dritten Schritt wird mit der SAS Base Funktion „probmc“ in Abhängigkeit 
von „Range“ ein q-Tabellenwert berechnet. Hiermit kann ein SNK-adjustierter p-
Wert für den Vergleich und ein Konfidenzintervall der Differenz berechnet wer-
den.  

%macro rename(factors); 
  %let k=1; 
  %let original = %scan(&factors, &k); 
     %do %while("&original" NE ""); 
      rename &original = _&original; 
        %let k = %eval(&k + 1); 
      %let original = %scan(&factors, &k); 
  %end; 
%mend; 
 
%macro critical_range_SNK(factors=, _factors=); 
*name of factors in lsmeans-Statement with and without underscore; 
 
*1st step: giving ranks to the means; 
*number of ranks depends on the number of means per effect-level; 
proc sort data=lsmeans_raw; 
  by effect estimate; run; 
proc rank data=lsmeans_raw out=lsmeans_raw_ranks1; 
  by effect; 
  var estimate;ranks rank1;run; 
data lsmeans_raw_ranks1; set lsmeans_raw_ranks1; 
  drop Estimate StdErr DF tValue Probt Alpha Lower Upper; run; 
*must drop some variables to avoid name conflict with lsmeans-file; 
proc rank data=lsmeans_raw out=lsmeans_raw_ranks2; 
  by effect; 
  var estimate;ranks rank2;run; 
data lsmeans_raw_ranks2; set lsmeans_raw_ranks2; 
  drop Estimate StdErr DF tValue Probt Alpha Lower Upper; run; 
*must drop some variables to avoid name conflict with lsmeans-file; 
proc sort data=lsmeans_raw_ranks1; 
  by effect &factors ; run; 
proc sort data=lsmeans_raw_ranks2; 
  by effect &factors ; run; 
data lsmeans_raw_ranks2; set lsmeans_raw_ranks2; 
  %rename(&factors); 
run; 
 
*2nd step: combine diffs-table with lsmeans-tables containing ranks; 
proc sort data=diffs_raw; 
  by effect &factors ; run; 
data diffs_raw2; merge diffs_raw lsmeans_raw_ranks1; 
  by effect &factors ; run; 
proc sort data=diffs_raw2; 
  by effect &_factors ; run; 
*rounding of 'range'  ('range' is difference of ranks plus 1); 
*if range = 1 then set to 2, otherwise no results for equal means; 
data diffs_raw3; merge diffs_raw2 lsmeans_raw_ranks2; 
  by effect &_factors ; 
range = ceil(abs(rank1-rank2)+1 
if range = . then delete; 
if range = 1 then range = 2; 
run; 

*3rd step calculate LSD and p-values with MC adjustment according to 
SNK-procedure; 
*use probmc function of SAS/Base; 
data diffs_raw3; set diffs_raw3; 
prob = 1-alpha; 
q_val = probmc("Range",.,prob,df,range); 
ProbSNK =  1-probmc("RANGE", sqrt(2)*abs(tvalue),.,df,range); 
CR_SNK =  q_val/sqrt(2)*stderr; 
run; 
data lsmeans_raw; set lsmeans_raw; run; 
data diffs_raw; set diffs_raw3; run; 
%mend; 
 
Tabelle 4: Mittelwertdifferenzen und p-Werte nach SNK-Adjustierung 

F1 _F1 Estimate Rank F1 Rank _F1 range prob q_val ProbSNK CR_SNK

1 2 -40.94 1 7 7 0.95 4.39 0.00 4.51
1 3 -41.14 1 9 9 0.95 4.64 0.00 4.76
2 3 -0.20 7 9 3 0.95 3.45 0.99 3.54
1 4 -41.13 1 8 8 0.95 4.53 0.00 4.65
2 4 -0.19 7 8 2 0.95 2.86 0.90 2.94
3 4 0.02 9 8 2 0.95 2.86 0.99 2.94

... ... ... ... ... ... ... ... ... ...
12 14 9.07 6 2 5 0.95 4.04 0.00 4.15
13 14 12.44 10 2 9 0.95 4.64 0.00 4.76

3.3 Buchstabendarstellung signifikanter Differenzen (%mult) 

In Tabelle 4 sind die nach dem SNK-Verfahren signifikanten Mittelwertdifferenzen 
durch eine Überschreitungswahrscheinlichkeit ProbSNK < 5% markiert. Diese adjus-
tierten p-Werte können in einem von Piepho [5], [6] entwickelten SAS-Makro %mult 
verwendet werden, um eine Mittelwerttabelle mit der sogenannten „Buchstabendarstel-
lung“ zu ergänzen (siehe Tabelle 5). 
 
*select effect of interest; 
data lsmeans; set lsmeans_raw; where effect="F1"; run;  
data diffs; set diffs_raw; where effect="F1"; run;    
%mult(trt=F1, alpha=0.05, p=probSNK); 
 
Wie wir in Tab. 5 sehen, kann es hierbei zu paradoxen Ergebnissen kommen, da sich 
z.B. Variante 12 und 2 signifikant unterscheiden, 12 und 4 dagegen nicht. Vergleichen 
wir mit den Grenzdifferenzen aus PROC GLM in Tab. 1 so stellen wir fest, dass die 
Buchstabendarstellung rechnerisch korrekt ist. Für benachbarte Mittelwerte gilt eine 
Grenzdifferenz von 2.9. Bei drei benachbarten Mittelwerten beträgt die kritische Diffe-
renz jedoch bereits 3.5 und diese Differenz wird vom Vergleich 12-4 nicht erreicht. Das 
SNK-Verfahren sieht aber eine nachträgliche Korrektur solcher Sonderfälle vor. 
 



Poster 

91 

Tabelle 5: Output %mult-Makro. 
Mittelwerte mit gleichen Buchstaben unterscheiden sich nicht signifikant. 

F1 LSMEAN COL1 COL2 COL3 COL4 COL5 COL6 
7 90.658 A B . . . .
6 90.333 A B . . . .
8 89.565 A B . . . .
5 89.480 A B . . . .
13 89.383 A B . . . .
3 89.355 A B . . . .
4 89.338 A B . . . .
2 89.153 A . . . . .
12 86.015 . B C . . .
11 84.885 . . C . . .
10 81.000 . . . D . .
9 80.155 . . . D . .
14 76.943 . . . . E .
1 48.213 . . . . . F

3.4 Einfügen zusätzlicher und Löschen redundanter Buchstaben 
(%snk_cleaner, %DeleteRedundantColumns) 

Entsprechend der Regel, dass Vergleiche nicht signifikant sind, wenn gleichzeitig Ver-
gleiche unter Beteiligung von noch kleineren oder noch größeren Mittelwerten nicht 
signifikant sind, werden vom Makro %snk_cleaner zusätzliche Buchstaben eingefügt, 
redundante Buchstabenspalten gelöscht und Buchstaben abschließend umbenannt. Ta-
belle 6 zeigt das Endergebnis. 

Tabelle 6: Korrigierte Buchstabendarstellung. 

F1 LSMEAN COL1 COL2 COL3 COL4 COL5 
7 90.658 A . . . . 
6 90.333 A . . . . 
8 89.565 A . . . . 
5 89.480 A . . . . 

13 89.383 A . . . . 
3 89.355 A . . . . 
4 89.338 A . . . . 
2 89.153 A . . . . 

12 86.015 A B . . . 
11 84.885 . B . . . 
10 81.000 . . C . . 
9 80.155 . . C . . 

14 76.943 . . . D . 
1 48.213 . . . . E 

4 Nutzung in der PIAF-Umgebung 
Die vorgestellten Makros wurden in ein PIAFStat-Verfahren „A-Bl SNK“ integriert und 
stehen zur Nutzung bereit. 
 
 

 
Abbildung 1: Versuchsdaten werden in PIAFStat eingelesen und über ‚SAS ausführen‘ 
wird Verfahrensauswahl  gestartet. 
 
 

 
Abbildung 2: Verfahren A-BL SNK wird ausgewählt. 

Abbildung 3: Zuordnung der Klassifikationsmerkmale. 
 
 

 
Abbildung 4: Zuordnung der Analysemerkmale. 

Abbildung 5: Wahl der Optionen für die Programm-Abarbeitung. 
 
 
Tabelle 7: Output mit Buchstabendarstellung. Gruppierung nach SNK-Test (α=0.05) 

Fungizid PG-Nr.
adjust 

Mittelw. COL1 COL2
Beiz_A + Beiz_B  + 1,8 Fungi_A + 1,2 Fungi_B 9 88.94 A .
Beiz_A + Beiz_B  + 1,8 Fungi_A 7 88.46 A .
Beiz_A + Beiz_B  + 0,75 Fungi_C + 0,75 Fungi_C 10 87.84 A .
Beiz_A + Beiz_B  + 1,8 Fungi_A 8 86.98 A B
Beiz_A + Beiz_B 6 84.66 A B
Beiz_A                 + 1,8 Fungi_A + 1,2 Fungi_B 4 82.92 A B
Beiz_A                 + 1,8 Fungi_A 2 81.68 A B
Beiz_A                 + 0,75 Fungi_C + 0,75 Fungi_C 5 80.88 A B
Beiz_A                 + 1,8 Fungi_A 3 79.16 . B
Unbehandelte Kontrolle 1 78.58 . B

